`symODE2`
A symbolic package under SageMath for the analysis of second-order ordinary differential equations with polynomial coefficients.

The User Manual

Tolga Birkandan*

Department of Physics, Istanbul Technical University, 34469 Istanbul, Turkey.

June 25, 2021

1 Introduction

`symODE2` is an open-source package for symbolic analysis of second-order ordinary differential equations with polynomial coefficients in the form

$$f_1(z)\frac{d^2y(z)}{dz^2}+f_2(z)\frac{dy(z)}{dz}+f_3(z)y(z)=0, \tag{1}$$

where the coefficient functions $f_{1,2,3}(z)$ are polynomials in the parameter z.

The approach is mainly based on the singularity structure of the equation and the routines are written under the open-source computer algebra system SageMath.

The code is able to obtain the singularity structure, indices and recurrence relations associated with the regular singular points, and symbolic solutions of the hypergeometric equation, Heun equation, and their confluent forms. For the mathematical details and methods, see [1].

The commercial systems Maple and Mathematica are also able to make symbolic analysis on the same types of equations along with their full numerical treatment. However, our aim is to provide an open-source symbolic calculator. Moreover, the numerical analysis of the hypergeometric type equations are defined in SageMath and the work is in progress for a numerical treatment of the Heun type equations under Python and SageMath [2].

The standard forms of the equations defined in our code are given in the appendix along with the Maple and Mathematica notations, and parameter correspondences.

2 The code structure of the symODE2 package

The symODE2 package is written under SageMath 9.1 using a laptop computer with Intel(R) Core(TM) i7-6500U CPU @ 2.50GHz and 8 GB memory. The operating system is Windows 10 Enterprise ver.1909.

*E-mail: birkandant@itu.edu.tr

The package consists of two main parts:

- `ode2analyzer.sage` for the general analysis and,
- `hypergeometric_heun.sage` for the symbolic solutions of the equations.

`hypergeometric_heun.sage` calls the routines defined in `ode2analyzer.sage` when needed.

We suggest the user put these two files in the same directory. The parts of the package and a sample session can be downloaded from the address [3]:

`https://github.com/tbirkandan/symODE2`

2.1 General analysis: ode2analyzer

The first part, `ode2analyzer` contains the routines,

- `find_singularities(diffeqn,y,z)`: Finds the singularity structure of the input ODE. The output is an array that involves the locations of the singularities, indices of the regular singularities and the ranks of the irregular singularities.
- `find_indices_recurrence(diffeqn,y,z,point,index,operation)`: Finds the indices and/or the recurrence relation with respect to a regular singular point using the θ-operator method [1].
- `ode_change_of_variable(diffeqn,y,z,newvar,transformation)`: Performs a change of variables.
- `normal_form_ode2(diffeqn,y,z)`: Finds the normal form of a second order ODE. For an input in the form given by the equation (1), the output is in the form

$$\frac{d^2w}{dz^2} + q'(z)w = 0. \tag{2}$$

  Please see [1] for notation and details.

Here, the input arguments are

- `diffeqn`: A 2nd order ODE with polynomial coefficients.
- `y`: The dependent variable of the ODE.
- `z`: The independent variable of the ODE.
- `point`: The regular singular point for which the indices and/or the recurrence relation will be found.
- `index`: The name of the parameter that will be used for the characteristic exponent.
- `operation`: The output of the routine (indices, recurrence, full (both)).
- `newvar`: The new variable.
- `transformation`: The transformation function that defines the change of variables.

2.2 Symbolic solutions of special ODEs: hypergeometric_heun

The second part, `hypergeometric_heun` contains the routines,

- `find_special_ode(diffeqn,y,z)`: Finds the type of the ODE using its singularity structure and solves it using the routines `find_.(.,.,.)` defined below.
- `ode_finder_bruteforce(diffeqn,y,z)`: Uses a change of variables list in order to bring the input ODE into a special form that is recognized in the package.
- `find_2F1(diffeqn,y,z)`: Solves a hypergeometric equation.
- `find_1F1(diffeqn,y,z)`: Solves a confluent hypergeometric equation.
- `find_HG(diffeqn,y,z)`: Solves a general Heun equation.
- `find_HC(diffeqn,y,z)`: Solves a (singly) confluent Heun equation.
- `find_HD(diffeqn,y,z)`: Solves a double confluent Heun equation.
- `find_HB(diffeqn,y,z)`: Solves a biconfluent Heun equation.
- `find_HT(diffeqn,y,z)`: Solves a triconfluent Heun equation.

The input arguments (`diffeqn,y,z`) are the same as defined in the first part.

Let us reset the SageMath session, load the file containing the routines and declare our functions,

```
# Start the session
reset()
load('ode2analyzer.sage')
#######################
u = function('u')(x)
u(x)=u
f1 = function('f1')(x)
f2 = function('f2')(x)
f3 = function('f3')(x)
#######################
```

Here, the line `u(x)=u` is written in order to avoid a "Deprecation Warning" as explained in page 162 of [4]. The variables and functions that define the hypergeometric equation are given as

```
# Hypergeometric equation:
var('a,b,c');
f1(x)=x*(1-x);
f2(x)=(c-(a+b+1)*x);
f3(x)=-a*b;

myeqn=f1(x)*diff(u(x),x,x)+f2(x)*diff(u(x),x)+f3(x)*u(x)
show(myeqn)
```

The hypergeometric equation as given in eq. (52) is thus defined and shown as

$$-abu(x) - (x-1)x\frac{\partial^2}{(\partial x)^2}u(x) - ((a+b+1)x - c)\frac{\partial}{\partial x}u(x) \tag{3}$$

The singularity structure of the equation is found by the command

```
# Singularity structure
mystructure=find_singularities(diffeqn=myeqn,y=u,z=x)
show(mystructure)
```

as

$$([1,0,+\infty],[[-a-b+c,0],[-c+1,0],[b,a]],[],[]) \tag{4}$$

There are four sub-arrays in the resulting array. The first one contains the locations of the regular singular points, the second one contains the indices (characteristic exponents) associated with the regular singular points (respectively), the third one contains the locations of the irregular singular points and the last array contains the ranks of the irregular singular points (respectively).

Here, the singularity structure of the general hypergeometric equation is found correctly [5]. We can run our code for an ODE with irregular singularities as in the following example. The double confluent Heun equation 74 can be defined as

```
# Double confluent Heun:
var('alpha,gamma,delta,epsilon,q');
f1(x)=1;
f2(x)=(delta/x)+(gamma/(x^2))+epsilon;
f3(x)=(alpha*x-q)/(x^2)

myeqn=f1(x)*diff(u(x),x,x)+f2(x)*diff(u(x),x)+f3(x)*u(x)
show(myeqn)
```

$$\left(\epsilon+\frac{\delta}{x}+\frac{\gamma}{x^2}\right)\frac{\partial}{\partial x}u\left(x\right)+\frac{(\alpha x-q)u\left(x\right)}{x^2}+\frac{\partial^2}{(\partial x)^2}u\left(x\right) \tag{5}$$

and the singularity structure can be found by the same command as

$$([],[],[0,+\infty],[1,1]) \tag{6}$$

Here, we find two rank-1 irregular singular points located at (0, ∞) [5]. The (singly) confluent Heun equation (66) case can be analyzed as

```
# (Singly) Confluent Heun:
var('alpha,gamma,delta,epsilon,q');
f1(x)=1;
f2(x)=(gamma/x)+(delta/(x-1))+epsilon;
f3(x)=(alpha*x-q)/(x*(x-1))

myeqn=f1(x)*diff(u(x),x,x)+f2(x)*diff(u(x),x)+f3(x)*u(x)
show(myeqn)

mystructure=find_singularities(diffeqn=myeqn,y=u,z=x)
show(mystructure)
```

$$\left(\epsilon+\frac{\delta}{x-1}+\frac{\gamma}{x}\right)\frac{\partial}{\partial x}u\left(x\right)+\frac{(\alpha x-q)u\left(x\right)}{(x-1)x}+\frac{\partial^2}{(\partial x)^2}u\left(x\right) \tag{7}$$

$$([1,0],[[-\delta+1,0],[-\gamma+1,0]],[+\infty],[1]) \tag{8}$$

with two regular singularities at (0, 1) and one irregular singularity at (∞) of rank-1 [5].

Another routine of our code is able to find the indices and the recurrence relation for a given regular singular point of a second-order ODE with polynomial coefficients. An input parameter `operation` can take the values: `"indices"`, `"recurrence"`, and `"full"`, yielding the corresponding results (`"full"` gives both the indices and the recurrence relation).

For the hypergeometric equation, around $x=0$ we have

```
# Hypergeometric equation
var('a,b,c');f1(x)=x*(1-x);f2(x)=(c-(a+b+1)*x);f3(x)=-a*b;
myeqn=f1(x)*diff(u(x),x,x)+f2(x)*diff(u(x),x)+f3(x)*u(x)

var('r')
myindicesandrecurrence=find_indices_recurrence(diffeqn=myeqn,y=u,z=x,point=0,index=r,
  operation="full")
myindices=find_indices_recurrence(myeqn,u,x,0,r,"indices")
myrecurrence=find_indices_recurrence(myeqn,u,x,0,r,"recurrence")

show(myindicesandrecurrence)
show(myindices)
show(myrecurrence)
```

Here, “r” is the parameter associated with the characteristic exponent. The results of the commands with parameters `"indices"` and `"recurrence"` are

$$[-c+1,0] \tag{9}$$

$$\begin{aligned}&\big((a-1)b+(a+b-2)n+n^2+(a+b+2\,n-2)r+r^2-a+1\big)C\,(n-1)\\&-\big((c-1)n+n^2+(c+2\,n-1)r+r^2\big)C\,(n)\end{aligned} \tag{10}$$

We obtain the recurrence relation corresponding to $r=0$ by substituting this value in the recurrence relation and solving it for $C(n)$. The result will be in the form $C(n)=(...)C(n-1)$ as expected, namely

```
# The recurrence relation
solve(myrecurrence==0,C(n))[0].subs(r=0).rhs().full_simplify().collect_common_factors
  ()
```

$$C(n)=\frac{\big((a-1)b+(a+b-2)n+n^2-a+1\big)C\,(n-1)}{(c+n-1)n} \tag{11}$$

which agrees with the one given in [6].

The hypergeometric function is implemented in SageMath as in the prototype `hypergeometric([a,b],[c],x)`. Therefore we can verify our recurrence relation graphically using the code below. The code adds new terms to the summation if the new term is greater than the “machine epsilon” of the system.

```
# Plot of 2F1
def hypfromrecurrence(a,b,c,x):
machineepsilon=abs((7./3)-(4./3)-1)
Cn=1
n=0
result=1
while True:
coeff=(a*b + (a + b)*n + n^2)*(Cn)/((c + 1)*n + n^2 + c)
newterm=coeff*(x^(n+1))
if abs(newterm)<machineepsilon:
break
else:
result=result+newterm
Cn=coeff
n=n+1
return result

# https://dlmf.nist.gov/15.3.F2.mag
a=5
```

```
b=-10
c=1

hypsage(x)=hypergeometric([a,b],[c],x)

p1=plot(hypsage,(x,-0.023,0.7),legend_label='SageMath',axes_labels=[r'$x$', r'$_2F_1$([5,-10],[1],x)'])
p2=plot(hypfromrecurrence(a,b,c,x),(x,-0.023,0.7),legend_label='Recurrence relation', plot_points=10, linestyle='', marker='.',color='red')
(p1+p2).show()
```

We take the parameters a, b, c such that our result coincides with the plot given in [5] as Figure (15.3.2). We present the result of the SageMath and our recurrence relation on the same plot in Figure (1).



Figure 1: Plot of the hypergeometric function found by our recurrence relation result and the internal command of SageMath.

We should note that finding a solution using the recurrence relation generally requires more effort than presented here. Methods such as analytic continuation should be carefully applied in order to deal with the circle of convergence of the series solution [7, 8].

For the general Heun equation we have,

```
# Heun equation
var('alpha,beta,gamma,delta,epsilon,q,a1');
epsilon=alpha+beta+1-gamma-delta;
f1(x)=1;f2(x)=(gamma/x)+(delta/(x-1))+(epsilon/(x-a1));f3(x)=(alpha*beta*x-q)/(x*(x-1)*(x-a1))

myeqn=f1(x)*diff(u(x),x,x)+f2(x)*diff(u(x),x)+f3(x)*u(x)

var('r')
myrecurrence=find_indices_recurrence(myeqn,u,x,0,r,"recurrence").subs(r=0)
show(myrecurrence)

# The recurrence relation given in http://dlmf.nist.gov/31.3.E3
# with an overall minus sign
Rn=a1*(n+1)*(n+gamma)
```

```
Pn=(n-1+alpha)*(n-1+beta)
Qn=n*((n-1+gamma)*(1+a1)+a1*delta+epsilon)
recurrenceDLMF=-(Rn*C(n+1)-(Qn+q)*C(n)+Pn*C(n-1))
# We need to shift n by 1 to check with the DLMF result
myrecurrence=myrecurrence.subs(n=n+1)

# Their difference should be zero
shouldbezero=(myrecurrence-recurrenceDLMF).expand().factor().full_simplify()
show(shouldbezero)
```

We obtain the recurrence relation as

$$\begin{aligned}&\left((a_1+1)n^2-(a_1-1)\delta-a_1\gamma+((a_1-1)\delta+a_1\gamma-3a_1+\alpha+\beta-2)n+2\,a_1-\alpha-\beta+q+1\right)C\left(n-1\right)\\&-\left((\alpha-2)\beta+(\alpha+\beta-4)n+n^2-2\,\alpha+4\right)C\left(n-2\right)-\left(a_1n^2+(a_1\gamma-a_1)n\right)C\left(n\right)\end{aligned} \tag{12}$$

We also enter the recurrence relation given in [5] with an overall minus sign. The difference of the two results is zero, verifying the result of our code.

We plot the series solution for the general Heun equation in Figure (2) using a similar code with an array of plots and compare it with the plot given in the Wolfram Blog post [9] to see that they are similar. The details of this analysis can be found in the sample SageMath session [3].



Figure 2: Plot of the general Heun function similar to the one given in [9].

3 Change of variables for a second order ODE

Using the `ode_change_of_variable` routine of our code, we can work with this transformation for a general 2nd order ODE. Please see [1] for details.

```
# Change of variables:
show('Change of variables:')
reset()
load('ode2analyzer.sage')
######################
u = function('u')(x)
u(x)=u
f1 = function('f1')(x)
f2 = function('f2')(x)
f3 = function('f3')(x)
######################
```

```
myeqn=f1(x)*diff(u(x),x,x)+f2(x)*diff(u(x),x)+f3(x)*u(x)
show(myeqn)
var('t') #new variable
myresult=ode_change_of_variable(diffeqn=myeqn,y=u,z=x,newvar=t,transformation=sqrt(t))
show(myresult)
```

which yields

$$myeqn = f_3(x)\,u(x) + f_2(x)\frac{\partial}{\partial x}u(x) + f_1(x)\frac{\partial^2}{(\partial x)^2}u(x) \tag{13}$$

$$myresult = 4\,t f_1\left(\sqrt{t}\right)\frac{\partial^2}{(\partial t)^2}u(t) + f_3\left(\sqrt{t}\right)u(t) + 2\left(\sqrt{t}f_2\left(\sqrt{t}\right) + f_1\left(\sqrt{t}\right)\right)\frac{\partial}{\partial t}u(t) \tag{14}$$

4 Hypergeometric and Heun-type equations

The results of the hypergeometric and confluent hypergeometric equations are numerically usable as these functions are defined in SageMath. However, the numerical solutions of the Heun-type functions are not defined. The numerical solutions of the general Heun and (singly) confluent Heun functions are defined by Motygin for GNU Octave/MATLAB [7, 8]. GNU Octave/MATLAB commands can be run in a SageMath session. However, this procedure is not straightforward and it is beyond the scope of this work. The method given in [10] can also be employed in order to obtain numerical results. An optimized implementation of Giscard and Tamar's method is the subject of a future work [2].

The outputs of the following examples are obtained using the following code structure,

```
# Standard equations
reset()
load('hypergeometric_heun.sage') #Routines for the symbolic solutions
#######################
u = function('u')(x)
u(x)=u
f1 = function('f1')(x)
f2 = function('f2')(x)
f3 = function('f3')(x)
#######################

# Definition of the variables, f1, f2, and f3.

myeqn=f1(x)*diff(u(x),x,x)+f2(x)*diff(u(x),x)+f3(x)*u(x)
show(myeqn)
mysingularities=find_singularities(diffeqn=myeqn,y=u,z=x)
show(mysingularities)
myresult=find_special_ode(diffeqn=myeqn,y=u,z=x)
```

The singularity structures and the solutions for the corresponding equations are found by the code as in the following sections.

4.1 Hypergeometric equation

$$-abu\left(x\right) - (x-1)x\frac{\partial^{2}}{(\partial x)^{2}}u\left(x\right) - ((a+b+1)x-c)\frac{\partial}{\partial x}u\left(x\right) \tag{15}$$

$$\left(\left[1,0,+\infty\right],\left[\left[-a-b+c,0\right],\left[-c+1,0\right],\left[b,a\right]\right],\left[\right],\left[\right]\right) \tag{16}$$

$$\frac{C_{2}x\,{}_{2}F_{1}\left(\begin{matrix}a-c+1,b-c+1\\-c+2\end{matrix};x\right)+C_{1}x^{c}\,{}_{2}F_{1}\left(\begin{matrix}a,b\\c\end{matrix};x\right)}{x^{c}} \tag{17}$$

The result is numerically usable as `hypergeometric([a,b],[c],x)` functions are defined within SageMath.

4.2 Confluent hypergeometric equation

$$-au\left(x\right) + (b-x)\frac{\partial}{\partial x}u\left(x\right) + x\frac{\partial^{2}}{(\partial x)^{2}}u\left(x\right) \tag{18}$$

$$\left(\left[0\right],\left[\left[-b+1,0\right]\right],\left[+\infty\right],\left[1\right]\right) \tag{19}$$

$$C_{1}M\left(a,b,x\right) + C_{2}U\left(a,b,x\right) \tag{20}$$

The result is numerically usable as `hypergeometric_M(a,b,x)` and `hypergeometric_U(a,b,x)` functions are defined within SageMath.

4.3 (General) Heun equation

$$-\left(\frac{\alpha+\beta-\delta-\gamma+1}{a_{1}-x} - \frac{\delta}{x-1} - \frac{\gamma}{x}\right)\frac{\partial}{\partial x}u\left(x\right) - \frac{(\alpha\beta x-q)u\left(x\right)}{(a_{1}-x)(x-1)x} + \frac{\partial^{2}}{(\partial x)^{2}}u\left(x\right) \tag{21}$$

$$\left(\left[a_{1},1,0,+\infty\right],\left[\left[-\alpha-\beta+\delta+\gamma,0\right],\left[-\delta+1,0\right],\left[-\gamma+1,0\right],\left[\beta,\alpha\right]\right],\left[\right],\left[\right]\right) \tag{22}$$

$$\begin{gathered}C_{2}x^{-\gamma+1}\mathrm{HG}\left(a_{1},-(a_{1}\delta+\alpha+\beta-\delta-\gamma+1)(\gamma-1)+q,\alpha-\gamma+1,\beta-\gamma+1,-\gamma+2,\delta,x\right)\\+C_{1}\mathrm{HG}\left(a_{1},q,\alpha,\beta,\gamma,\delta,x\right)\end{gathered} \tag{23}$$

4.4 (Singly) Confluent Heun equation

$$\left(\frac{\nu}{x-1}+\frac{\mu}{x}\right)u\left(x\right) + \left(\alpha+\frac{\gamma+1}{x-1}+\frac{\beta+1}{x}\right)\frac{\partial}{\partial x}u\left(x\right) + \frac{\partial^{2}}{(\partial x)^{2}}u\left(x\right) \tag{24}$$

$$\left(\left[1,0\right],\left[\left[-\gamma,0\right],\left[-\beta,0\right]\right],\left[+\infty\right],\left[1\right]\right) \tag{25}$$

$$\begin{gathered}C_{1}\mathrm{HC}\left(\alpha,\beta,\gamma,-\frac{1}{2}\,\alpha\beta-\frac{1}{2}\,\alpha\gamma-\alpha+\mu+\nu,\frac{1}{2}\,(\alpha-1)\beta-\frac{1}{2}\,(\beta+1)\gamma+\frac{1}{2}\,\alpha-\mu,x\right)\\+\frac{C_{2}\mathrm{HC}\left(\alpha,-\beta,\gamma,-\frac{1}{2}\,\alpha\beta-\frac{1}{2}\,\alpha\gamma-\alpha+\mu+\nu,\frac{1}{2}\,(\alpha-1)\beta-\frac{1}{2}\,(\beta+1)\gamma+\frac{1}{2}\,\alpha-\mu,x\right)}{x^{\beta}}\end{gathered} \tag{26}$$

4.5 Double confluent Heun equation

$$\left(\epsilon+\frac{\delta}{x}+\frac{\gamma}{x^{2}}\right)\frac{\partial}{\partial x}u\left(x\right) + \frac{(\alpha x-q)u\left(x\right)}{x^{2}} + \frac{\partial^{2}}{(\partial x)^{2}}u\left(x\right) \tag{27}$$

$$\left(\left[\right],\left[\right],\left[0,+\infty\right],\left[1,1\right]\right) \tag{28}$$

$$C_{2}x^{-\delta+2}\mathrm{HD}\left(\delta+q-2,\alpha-2\,\epsilon,-\gamma,-\delta+4,-\epsilon,x\right)e^{\left(-\epsilon x+\frac{\gamma}{x}\right)} + C_{1}\mathrm{HD}\left(q,\alpha,\gamma,\delta,\epsilon,x\right) \tag{29}$$

4.6 Biconfluent Heun equation

$$\left(\epsilon x+\delta+\frac{\gamma}{x}\right)\frac{\partial}{\partial x}u\left(x\right)+\frac{(\alpha x-q)u\left(x\right)}{x}+\frac{\partial^2}{(\partial x)^2}u\left(x\right) \tag{30}$$

$$\left(\left[0\right],\left[\left[-\gamma+1,0\right]\right],\left[+\infty\right],\left[2\right]\right) \tag{31}$$

$$C_2 x^{-\gamma+1}\mathrm{HB}\left(\delta(\gamma-1)+q,-\epsilon(\gamma-1)+\alpha,-\gamma+2,\delta,\epsilon,x\right)+C_1\mathrm{HB}\left(q,\alpha,\gamma,\delta,\epsilon,x\right) \tag{32}$$

4.7 Triconfluent Heun equation

$$(\alpha x-q)u\left(x\right)+\left(\epsilon x^2+\delta x+\gamma\right)\frac{\partial}{\partial x}u\left(x\right)+\frac{\partial^2}{(\partial x)^2}u\left(x\right) \tag{33}$$

$$\left(\left[\right],\left[\right],\left[+\infty\right],\left[3\right]\right) \tag{34}$$

$$C_2\mathrm{HT}\left(\delta+q,\alpha-2\,\epsilon,-\gamma,-\delta,-\epsilon,x\right)e^{\left(-\frac{1}{3}\,\epsilon x^3-\frac{1}{2}\,\delta x^2-\gamma x\right)}+C_1\mathrm{HT}\left(q,\alpha,\gamma,\delta,\epsilon,x\right) \tag{35}$$

5 Applications with symODE2

We will study some examples in the literature in order to test our code. For further applications, the user should keep in mind that some transformations or identities may be required to obtain the same result with the literature.

5.1 Hypergeometric equation: Equation (15) of [11]

The equation (15) of Nasheeha et. al [11],

$$4\left(ax+1\right)(bx+1)\frac{\partial^2}{(\partial x)^2}u\left(x\right)-\left(\left(a(2\,\alpha+1)^2-2\,a+b\right)b+ab-b^2\right)u\left(x\right)+2\left(a-b\right)\frac{\partial}{\partial x}u\left(x\right) \tag{36}$$

can be solved using our code,

```
# The equation (15) of Nasheeha et. al
reset()
load('hypergeometric_heun.sage') #Routines for the symbolic solutions
########################
u = function('u')(x)
u(x)=u
f1 = function('f1')(x)
f2 = function('f2')(x)
f3 = function('f3')(x)
########################
var('a,b,k,alpha');
f1(x)=4*(1+a*x)*(1+b*x);
f2(x)=2*(a-b);
f3(x)=((b^2)-a*b-k*b)

myeqn=(f1(x)*diff(u(x),x,x)+f2(x)*diff(u(x),x)+f3(x)*u(x))
myresult=find_special_ode(diffeqn=myeqn,y=u,z=x)
```

as

$$\frac{\sqrt{ax+1}C_2\sqrt{b}\,{}_2F_1\left(\begin{matrix}-\alpha-\frac{1}{2},\alpha+\frac{1}{2}\\ \frac{3}{2}\end{matrix};-\frac{abx+b}{a-b}\right)+C_1\sqrt{-a+b}\,{}_2F_1\left(\begin{matrix}-\alpha-1,\alpha\\ \frac{1}{2}\end{matrix};-\frac{abx+b}{a-b}\right)}{\sqrt{-a+b}} \tag{37}$$

Here, the code found the change of variables itself. If we use the transformation given in the article, we find the same result as in the article, namely,

```
# Change of variables
var('t')
trnsf=((a-b)*t-a)/(a*b)
myeqn2=ode_change_of_variable(myeqn,u,x,t,trnsf)
myeqn2=myeqn2.subs(k==a*(2*alpha+1)^2-2*a+b)
myresult=find_special_ode(myeqn2,u,t)
show(myresult)
```

and

$$C_2 t^{\frac{3}{2}}\, {}_2F_1\left(\begin{matrix}-\alpha+\frac{1}{2}, \alpha+\frac{3}{2} \\ \frac{5}{2}\end{matrix}; t\right) + C_1\, {}_2F_1\left(\begin{matrix}-\alpha-1, \alpha \\ -\frac{1}{2}\end{matrix}; t\right) \tag{38}$$

5.2 Confluent hypergeometric equation: Whittaker equation

The Whittaker equation [5],

$$\frac{\mathrm{d}^2 u}{\mathrm{d}z^2} + \left(-\frac{1}{4} + \frac{\kappa}{x} + \frac{\frac{1}{4}-\mu^2}{x^2}\right) u = 0, \tag{39}$$

can be obtained from the confluent hypergeometric equation. Hence the solutions can be given in terms of the confluent hypergeometric functions M and U. We start by the substitution $u(x) = e^{-x/2}u(x)$ and the code

```
# The Whittaker equation
var('kappa,mu');f1(x)=1;f2(x)=0;f3(x)=(-1/4)+(kappa/x)+((1/4)-(mu^2))/(x^2)
myeqn=f1(x)*diff(u(x),x,x)+f2(x)*diff(u(x),x)+f3(x)*u(x)
myeqn=myeqn.substitute_function(u(x),exp(-x/2)*u(x))
myresult=find_special_ode(diffeqn=myeqn,y=u,z=x)
```

yields

$$\left(C_1 M\left(-\kappa+\mu+\frac{1}{2}, 2\,\mu+1, x\right) + C_2 U\left(-\kappa+\mu+\frac{1}{2}, 2\,\mu+1, x\right)\right) x^{\mu+\frac{1}{2}} \tag{40}$$

as the same result given in [5]. The code finds the factor $x^{\mu+\frac{1}{2}}$ itself.

5.3 General Heun equation: Equation (10) of [12]

The equation (10) of Petroff [12] is

$$-\left(x^2-1\right)(A-x)\frac{\partial^2}{(\partial x)^2}u\left(x\right) - 4\,Au\left(x\right) - \left(5\,Ax - 4\,x^2 - 1\right)\frac{\partial}{\partial x}u\left(x\right) \tag{41}$$

The author applies a change of variables as $z = (1-x)/2$. Let us use this transformation and solve the resulting equation with our code,

```
# General Heun equation: Equation (10) of Petroff:
var('A');f1(x)=(x-A)*((x^2)-1);f2(x)=4*(x^2)-5*A*x+1;f3(x)=-4*A;
myeqn=(f1(x)*diff(u(x),x,x)+f2(x)*diff(u(x),x)+f3(x)*u(x))
show(myeqn)
var('z')
myeqn2=ode_change_of_variable(diffeqn=myeqn,y=u,z=x,newvar=z,transformation=1-2*z)
myresult=find_special_ode(diffeqn=myeqn2,y=u,z=z)
show(myresult)
```

that yields

$$C_1 \mathrm{HG}\left(-\frac{1}{2}\,A+\frac{1}{2},-2\,A,3,0,\frac{5}{2},\frac{5}{2},z\right)+\frac{C_2 \mathrm{HG}\left(-\frac{1}{2}\,A+\frac{1}{2},-\frac{1}{8}\,A-\frac{3}{8},\frac{3}{2},-\frac{3}{2},-\frac{1}{2},\frac{5}{2},z\right)}{z^{\frac{3}{2}}} \tag{42}$$

which is the same with the reference.

5.4 Confluent Heun equation: Equation (26) of [13]

The ODE (26) in Sakallı et al. [13],

$$-(x-1)x\frac{\partial^2}{(\partial x)^2}u(x)-\left(L_1 m^2 r(x-1)-L_1^2\omega^2\left(\frac{1}{x}-1\right)-L_1(L_2+1)L_2\right)u(x)-(2\,x-1)\frac{\partial}{\partial x}u(x) \tag{43}$$

needs the introduction of the factor $x^{i\,L_1\omega}$ given in eq. (27) of this article in order to obtain it in the standard form of the (singly) confluent Heun equation. Our code finds this function automatically. Keeping in mind that L_1 is L and L_2 is ℓ, we run

```
# The ODE (26) in Sakalli et al.
var('L1,L2,omega,m,r');
f1(x)=x*(1-x);
f2(x)=1-2*x;
f3(x)=((L1^2)*(omega^2)*((1/x)-1))+((m^2)*(L1)*r*(1-x))+(L2*(L2+1)*L1)

myeqn=(f1(x)*diff(u(x),x,x)+f2(x)*diff(u(x),x)+f3(x)*u(x))
myresult=find_special_ode(diffeqn=myeqn,y=u,z=x)
```

and we get

$$\begin{aligned}&C_2 x^{i\,L_1\omega}x^{-2i\,L_1\omega}\mathrm{HC}\left(0,-2i\,L_1\omega,0,L_1 m^2 r,-L_1 m^2 r-L_1 L_2^2-L_1 L_2,x\right)\\&+C_1 x^{i\,L_1\omega}\mathrm{HC}\left(0,2i\,L_1\omega,0,L_1 m^2 r,-L_1 m^2 r-L_1 L_2^2-L_1 L_2,x\right)\end{aligned} \tag{44}$$

which agrees with the original article.

5.5 Biconfluent Heun equation: Equation (10) of [14]

The equation (10) of Vitória et. al [14] can be solved in terms of biconfluent Heun functions. We need the substitution $u(x)\to x^\gamma e^{-\frac{x^2}{2}}u(x)$ and the transformation $x=-z$. Using the code

```
# The equation (10) of Vitoria et. al
var('gamma,beta,m,omega,delta')
f1(x)=1
f2(x)=1/x
f3(x)=(beta/(m*omega))+(delta/x)-((gamma^2)/(x^2))-x^2
myeqn=f1(x)*diff(u(x),x,x)+f2(x)*diff(u(x),x)+f3(x)*u(x)
myeqn=myeqn.substitute_function(u(x),exp(-(x^2)/2)*(x^gamma)*u(x))
var('z')
myeqn=ode_change_of_variable(diffeqn=myeqn,y=u,z=x,newvar=z,transformation=-z)
show(myeqn)
myresult=find_special_ode(diffeqn=myeqn,y=u,z=z)
show(myresult)
```

we obtain the equation,

$$\frac{\frac{\partial}{\partial z}u(z)}{z}-\frac{\left(m\omega z^4+\gamma^2 m\omega+\delta m\omega z-\beta z^2\right)u(z)}{m\omega z^2}+\frac{\partial^2}{(\partial z)^2}u(z) \tag{45}$$

and the solution,

$$C_1 \mathrm{HB}\left(\delta, -2\,\gamma + \frac{\beta}{m\omega} - 2, 2\,\gamma + 1, 0, -2, z\right) e^{\left(-\frac{1}{2}\,z^2\right)} + \frac{C_2 \mathrm{HB}\left(\delta, 2\,\gamma + \frac{\beta}{m\omega} - 2, -2\,\gamma + 1, 0, -2, z\right) e^{\left(-\frac{1}{2}\,z^2\right)}}{z^{2\,\gamma}} \tag{46}$$

as given in the original article after employing the correspondence given in equations (78)-(82).

5.6 Double confluent Heun equation: Equation (11) of [15]

As explained in the appendix, the singularity structure of the Maple-based approach of the double confluent Heun equation is different from the Mathematica version. The symODE2 package uses the Mathematica form as it involves more parameters. However, this choice makes it difficult to verify our analysis with the examples in literature.

The equation (11) of Vieira [15],

$$\left(A_1 + \frac{A_2}{x^2} + \frac{A_3}{x^4}\right) u\left(x\right) + \frac{\frac{\partial}{\partial x} u\left(x\right)}{x} + \frac{\partial^2}{(\partial x)^2} u\left(x\right) \tag{47}$$

has the singularity structure $([]\,, []\,, [0, +\infty]\,, [1, 1])$. The equation can be solved with the code

```
# The equation (11) of Vieira
var('A1,A2,A3');f1(x)=1;f2(x)=1/x;f3(x)=A1+(A2/x^2)+(A3/x^4)

myeqn=f1(x)*diff(u(x),x,x)+f2(x)*diff(u(x),x)+f3(x)*u(x)
myresult=find_special_ode(diffeqn=myeqn,y=u,z=x)
show(myresult)
```

as

$$\begin{aligned} &C_2 \mathrm{HD}\left(-A_2 - 2\,\sqrt{-A_1}\sqrt{-A_3} - \frac{1}{4}, -2\,\sqrt{-A_1}, -2\,\sqrt{-A_3}, 2, -2\,\sqrt{-A_1}, x\right) e^{\left(\frac{1}{3}\,\sqrt{-A_1}x^3 + \frac{1}{4}\,x^2 - 2\,\sqrt{-A_1}x + \sqrt{-A_3}x + \frac{2\,\sqrt{-A_3}}{x}\right)} \\ &+ C_1 \mathrm{HD}\left(-A_2 - 2\,\sqrt{-A_1}\sqrt{-A_3} - \frac{1}{4}, 2\,\sqrt{-A_1}, 2\,\sqrt{-A_3}, 2, 2\,\sqrt{-A_1}, x\right) e^{\left(\frac{1}{3}\,\sqrt{-A_1}x^3 + \frac{1}{4}\,x^2 + \sqrt{-A_3}x\right)} \end{aligned} \tag{48}$$

If we apply the transformation (12) $\left(t \to \frac{x^2+1}{x^2-1}\right)$ as given in the article, we find the form of the equation (13) in the article that can be related with the Maple form, namely,

$$\frac{(A_1 + A_2 + A_3)t^2 + 2\,(A_1 - A_3)t + A_1 - A_2 + A_3}{t^6 - 3\,t^4 + 3\,t^2 - 1} u\left(t\right) + \frac{2\left(t^5 - 2\,t^3 + t\right)}{(t+1)^3 (t-1)^3} \frac{\partial}{\partial t} u\left(t\right) + \frac{\partial^2}{(\partial t)^2} u\left(t\right) \tag{49}$$

The singularity structure of this equation is $([]\,, []\,, [-1, 1]\,, [1, 1])$ as used by Maple.

5.7 Triconfluent Heun equation: Equation (5) of [16]

The equation (5) of Dong et. al [16],

$$-\frac{\sqrt{2}\left(2\,bx^2 + a\right) \frac{\partial}{\partial x} u\left(x\right)}{\sqrt{b}} + \frac{\left(a^2 + 4\,b\epsilon - 4\,\sqrt{2}\sqrt{b^3}x\right) u\left(x\right)}{2\,b} + \frac{\partial^2}{(\partial x)^2} u\left(x\right) \tag{50}$$

needs the change of coordinates $z = \sqrt[6]{\frac{8b}{9}}x$ in order to bring it in a specific form of the triconfluent Heun equation. Let us apply this transformation and solve the resulting equation with the code

```
# The equation (5) of Dong et. al
var('a,b,epsilon');
f1(x)=1
```

```
f2(x)=-sqrt(2)*(a+2*b*x^2)/sqrt(b)
f3(x)=((a^2)-4*sqrt(2*b^3)*x+4*b*epsilon)/(2*b)
myeqn=(f1(x)*diff(u(x),x,x)+f2(x)*diff(u(x),x)+f3(x)*u(x))
show(myeqn)
show(find_singularities(diffeqn=myeqn,y=u,z=x))
var('z')
myeqn2=ode_change_of_variable(diffeqn=myeqn,y=u,z=x,newvar=z,transformation=z*((8*b)/9)^-(1/6))
myresult=find_special_ode(diffeqn=myeqn2,y=u,z=z)
show(myresult)
```

which yields

$$C_1 \mathrm{HT}\left(-\frac{3^{\frac{2}{3}}a^2}{4\,b^{\frac{4}{3}}}-\frac{3^{\frac{2}{3}}\epsilon}{b^{\frac{1}{3}}},3,\frac{3^{\frac{1}{3}}a}{b^{\frac{2}{3}}},0,3,z\right)e^{\left(z^3+\frac{3^{\frac{1}{3}}az}{b^{\frac{2}{3}}}\right)}+C_2 \mathrm{HT}\left(-\frac{3^{\frac{2}{3}}a^2}{4\,b^{\frac{4}{3}}}-\frac{3^{\frac{2}{3}}\epsilon}{b^{\frac{1}{3}}},-3,-\frac{3^{\frac{1}{3}}a}{b^{\frac{2}{3}}},0,-3,z\right) \tag{51}$$

that agrees with the original paper using the correspondence of the parameters given in equations (86-90).

5.8 Further applications

The analysis of other ODEs will be in the same manner, therefore we will just give a list of some other cases. The corresponding analyses can be found in the sample SageMath session given in [3].

- General Heun: Lamé equation as given in [17].
- Biconfluent Heun: The quantum-mechanical doubly anharmonic oscillator potential as given in [18].
- Triconfluent Heun: The classical anharmonic oscillator equation as given in [19]. It should be noted that our code finds the parameter ϵ using a quadratic equation favoring the positive root. The Mathematica solution has the negative root.
- Confluent Heun: Equation (57) of [20].
- Hypergeometric: Legendre equation as given in [5].
- Confluent hypergeometric: Hermite equation as given in [21]. The equations (7.2) and (7.3) of this paper can be obtained from the solution of the Hermite equation given in the same paper. The identity (13.8(iii)) in [5] should be used in the calculation.

A Appendix: Hypergeometric and Heun type equations in DLMF, Maple, Mathematica, and symODE2

The standard forms of the equations may be defined differently in the literature and in the computer algebra systems. Here, we consider DLMF, a well-known library of mathematical functions [5], and two computer algebra systems, Maple and Mathematica which can work with hypergeometric and Heun type functions. We give lists of correspondence of the parameters in these programs. We also note the standard forms of the equations used in the symODE2 code.

A.1 Hypergeometric equation

In DLMF [5],

$$x(1-x)\frac{d^2y}{dx^2}+[c-(a+b+1)y]\frac{dy}{dx}-aby=0. \tag{52}$$

In Maple [22],

$$x(x-1)\frac{d^2y}{dx^2}+[(a+b+1)x-c]\frac{dy}{dx}+aby=0, \tag{53}$$

with solution `hypergeom([a,b],[c],x)`.

In Mathematica [23],

$$-\left(x(x-1)\frac{d^2y}{dx^2}+[(a+b+1)x-c]\frac{dy}{dx}+aby\right)=0, \tag{54}$$

with solution `Hypergeometric2F1[a,b,c,z]`.

All equations coincide and they have three regular singularities located at $\{0,\,1,\,\infty\}$. symODE2 uses this form of the equation as well.

A.2 Confluent hypergeometric equation

In DLMF [5],

$$x\frac{d^2y}{dx^2}+(b-x)\frac{dy}{dx}-ay=0. \tag{55}$$

In Maple [22],

$$x\frac{d^2y}{dx^2}+(c-x)\frac{dy}{dx}-ay=0, \tag{56}$$

with solution `hypergeom([a],[c],x)`.

In Mathematica [24],

$$x\frac{d^2y}{dx^2}+(b-x)\frac{dy}{dx}-ay=0, \tag{57}$$

with solution `HypergeometricU[a,b,x]`.

All have one regular singularity located at $\{0\}$ and one irregular singularity of rank-1 at $\{\infty\}$. The equations coincide and symODE2 also uses this form of the equation.

A.3 (General) Heun equation

In DLMF [5],

$$\frac{d^2y}{dx^2}+\left(\frac{\gamma}{x}+\frac{\delta}{x-1}+\frac{\epsilon}{x-a}\right)\frac{dy}{dx}+\frac{\alpha\beta x-q}{x(x-1)(x-a)}y=0, \tag{58}$$

and in Maple [25],

$$\frac{d^2y}{dx^2}+\left(\frac{\gamma}{x}+\frac{\delta}{x-1}+\frac{\epsilon}{x-a}\right)\frac{dy}{dx}+\frac{\alpha\beta x-q}{x(x-1)(x-a)}y=0, \tag{59}$$

where $\epsilon=\alpha+\beta+1-\gamma-\delta$ with solution $\texttt{HeunG}(a,q,\alpha,\beta,\gamma,\delta,x)$.

In Mathematica [26],

$$\frac{d^2y}{dx^2}+\left(\frac{\gamma}{x}+\frac{\delta}{x-1}+\frac{\alpha+\beta+1-\gamma-\delta}{x-a}\right)\frac{dy}{dx}+\frac{\alpha\beta x-q}{x(x-1)(x-a)}y=0, \tag{60}$$

with solution $\texttt{HeunG}(a,q,\alpha,\beta,\gamma,\delta,x)$.

All have four regular singularities located at $\{0,\,1,\,a,\,\infty\}$. symODE2 uses the same structure of the equation.

A.4 (Singly) Confluent Heun equation

In DLMF [5],

$$\frac{d^2y}{dx^2} + \left(\frac{\gamma}{x} + \frac{\delta}{x-1} + \epsilon\right)\frac{dy}{dx} + \frac{\alpha x - q}{x(x-1)}y = 0. \tag{61}$$

In Maple [25],

$$\begin{aligned}&\frac{d^2y}{dx^2} - \frac{-x^2\alpha + (-\beta+\alpha-\gamma-2)\,x + \beta + 1}{x\,(x-1)}\frac{dy}{dx}\\ &-\frac{[(-\beta-\gamma-2)\,\alpha - 2\,\delta]\,x + (\beta+1)\,\alpha + (-\gamma-1)\,\beta - 2\,\eta - \gamma}{2x\,(x-1)}y = 0,\end{aligned} \tag{62}$$

with solution $\texttt{HeunC}(\alpha, \beta, \gamma, \delta, \eta, x)$. This form can be transformed into

$$\frac{d^2y}{dx^2} + \left(\frac{\beta+1}{x} + \frac{\gamma+1}{x-1} + \alpha\right)\frac{dy}{dx} + \left(\frac{\mu}{x} + \frac{\nu}{x-1}\right)y = 0, \tag{63}$$

where

$$\delta = \mu + \nu - \alpha\frac{\beta+\gamma+2}{2}, \tag{64}$$

$$\eta = \frac{(\alpha-\gamma)(\beta+1)-\beta}{2} - \mu. \tag{65}$$

In Mathematica [26],

$$\frac{d^2y}{dx^2} + \left(\frac{\gamma}{x} + \frac{\delta}{x-1} + \epsilon\right)\frac{dy}{dx} + \frac{\alpha x - q}{x(x-1)}y = 0, \tag{66}$$

with solution $\texttt{HeunC}(q, \alpha, \gamma, \delta, \epsilon, x)$.

We can define the correspondence,

$$\gamma_{Mat} = \beta_{Map} + 1, \tag{67}$$

$$\delta_{Mat} = \gamma_{Map} + 1, \tag{68}$$

$$\epsilon_{Mat} = \alpha_{Map}, \tag{69}$$

$$\alpha_{Mat} = \mu_{Map} + \nu_{Map}, \tag{70}$$

$$q_{Mat} = \mu_{Map}, \tag{71}$$

where "Mat" and "Map" stand for Mathematica and Maple, respectively.

All have two regular singularities located at {0,1} and an irregular singularity of rank-1 at $\{\infty\}$. symODE2 uses the Maple form.

A.5 Double confluent Heun equation

In DLMF [5],

$$\frac{d^2y}{dx^2} + \left(\frac{\delta}{x^2} + \frac{\gamma}{x} + 1\right)\frac{dy}{dx} + \frac{\alpha x - q}{x^2}y = 0. \tag{72}$$

This equation has two irregular singular points of rank-1 located at $\{0, \infty\}$.

In Maple [25],

$$\frac{d^2y}{dx^2} - \frac{\alpha\,x^4 - 2\,x^5 + 4\,x^3 - \alpha - 2\,x}{(x-1)^3\,(x+1)^3}\frac{dy}{dx} - \frac{-x^2\beta + (-\gamma-2\,\alpha)\,x - \delta}{(x-1)^3\,(x+1)^3}y = 0, \tag{73}$$

with solution $\texttt{HeunD}(\alpha, \beta, \gamma, \delta, x)$. This equation has two irregular singular points of rank-1 located at {-1, 1}.

In Mathematica [26],

$$\frac{d^2y}{dx^2} + \left(\frac{\gamma}{x^2} + \frac{\delta}{x} + \epsilon\right)\frac{dy}{dx} + \frac{\alpha x - q}{x^2}y = 0, \tag{74}$$

with solution HeunD$(q, \alpha, \gamma, \delta, \epsilon, x)$. This equation has two irregular singular points of rank-1 located at $\{0,\ \infty\}$. symODE2 uses the Mathematica form.

We will not write a correspondence between the Maple and Mathematica parameters as the singularity structures of the equations are not the same.

A.6 Biconfluent Heun equation

In DLMF [5],

$$\frac{d^2y}{dx^2} - \left(\frac{\gamma}{x} + \delta + x\right)\frac{dy}{dx} + \frac{\alpha x - q}{z}y = 0. \tag{75}$$

In Maple [25],

$$\frac{d^2y}{dx^2} - \frac{\beta\, x + 2\, x^2 - \alpha - 1}{x}\frac{dy}{dx} - \frac{(2\,\alpha - 2\,\gamma + 4)\, x + \beta\,\alpha + \beta + \delta}{2x}y = 0, \tag{76}$$

with solution HeunB$(\alpha, \beta, \gamma, \delta, x)$.

In Mathematica [26],

$$\frac{d^2y}{dx^2} + \left(\frac{\gamma}{x} + \delta + \epsilon x\right)\frac{dy}{dx} + \frac{\alpha x - q}{x}y = 0, \tag{77}$$

with solution HeunB$(q, \alpha, \gamma, \delta, \epsilon, x)$.

All have one regular singularity located at $\{0\}$ and one irregular singularity of rank-2 at $\{\infty\}$. symODE2 uses the Mathematica form.

We can define the correspondence,

$$\begin{aligned}
\gamma_{Mat} &= \alpha_{Map} + 1, && (78)\\
\delta_{Mat} &= -\beta_{Map}, && (79)\\
\epsilon_{Mat} &= -2, && (80)\\
\alpha_{Mat} &= \gamma_{Map} - \alpha_{Map} - 2, && (81)\\
q_{Mat} &= \frac{\beta_{Map}(\alpha_{Map} + 1) + \delta_{Map}}{2}, && (82)
\end{aligned}$$

where "Mat" and "Map" stand for Mathematica and Maple, respectively.

A.7 Triconfluent Heun equation

In DLMF [5],

$$\frac{d^2y}{dx^2} + (\gamma + x)\, x\frac{dy}{dx} + (\alpha x - q)\, y = 0. \tag{83}$$

In Maple [25],

$$\frac{d^2y}{dx^2} - (3x^2 + \gamma)\frac{dy}{dx} + [(\beta - 3)\, x + \alpha]y = 0, \tag{84}$$

with solution HeunT$(\alpha, \beta, \gamma, x)$.

In Mathematica [26],

$$\frac{d^2y}{dx^2} + (\gamma + \delta x + \epsilon x^2)\frac{dy}{dx} + (\alpha x - q)y = 0, \tag{85}$$

with solution HeunT$(q, \alpha, \gamma, \delta, \epsilon, x)$.

All have one irregular singularity of rank-3 at $\{\infty\}$. symODE2 uses the Mathematica form.

We can define the correspondence,

$$\begin{aligned}
\epsilon_{Mat} &= -3, && (86)\\
\delta_{Mat} &= 0, && (87)\\
\gamma_{Mat} &= -\gamma_{Map}, && (88)\\
\alpha_{Mat} &= \beta_{Map} - 3, && (89)\\
q_{Mat} &= -\alpha_{Map}, && (90)
\end{aligned}$$

where "Mat" and "Map" stand for Mathematica and Maple, respectively.

References

[1] T. Birkandan, "Symbolic analysis of second-order ordinary differential equations with polynomial coefficients", arXiv:2010.01563 [math-ph].

[2] T. Birkandan, P.-L. Giscard and A. Tamar, "Python computations of general Heun functions from their integral series representations", in progress.

[3] "The symODE2 package", Accessed 2 March 2021, https://github.com/tbirkandan/symODE2

[4] R.A. Mezei, *An Introduction to SAGE Programming*, Wiley, Hoboken (2016).

[5] F.W.J. Olver, A.B. Olde Daalhuis, D.W. Lozier, B.I. Schneider, R.F. Boisvert, C.W. Clark, B.R. Miller, B.V. Saunders, H.S. Cohl, and M.A. McClain (eds.), *NIST Digital Library of Mathematical Functions*, http://dlmf.nist.gov/, Release 1.0.27 of 2020-06-15.

[6] S.Yu. Slavyanov and W. Lay, *Special Functions, A Unified Theory Based on Singularities*, Oxford University Press, New York (2000).

[7] O.V. Motygin, "On numerical evaluation of the Heun functions", Proceedings of the 2015 Days on Diffraction (DD), IEEE (2015). [arXiv:1506.03848 [math.NA]].

[8] O.V. Motygin, "On evaluation of the confluent Heun functions", Proceedings of the 2018 Days on Diffraction (DD), IEEE (2018). [arXiv:1804.01007 [math.NA]].

[9] T. Ishkhanyan, "From Sine to Heun: 5 New Functions for Mathematics and Physics in the Wolfram Language", Accessed 13 September 2020, https://blog.wolfram.com/2020/05/06/from-sine-to-heun-5-new-functions-for-mathematics-and-physics-in-the-wolfram-language/

[10] P.-L. Giscard and A. Tamar, "Elementary Integral Series for Heun Functions. With an Application to Black-Hole Perturbation Theory," arXiv:2010.03919 [math-ph] (2020).

[11] R.N. Nasheeha, S. Thirukkanesh and F.C. Ragel, "Anisotropic generalization of isotropic models via hypergeometric equation," Eur. Phys. J. C **80**, no.1, 6 (2020).

[12] D. Petroff, "Slowly rotating homogeneous stars and the Heun equation," Class. Quant. Grav. **24**, 1055-1068 (2007). [arXiv:gr-qc/0701081 [gr-qc]].

[13] İ. Sakallı, K. Jusufi and A. Övgün, "Analytical Solutions in a Cosmic String Born-Infeld-dilaton Black Hole Geometry: Quasinormal Modes and Quantization," Gen. Rel. Grav. **50**, no.10, 125 (2018). [arXiv:1803.10583 [gr-qc]].

[14] R.L.L. Vitória, C. Furtado and K. Bakke, "On a relativistic particle and a relativistic position-dependent mass particle subject to the Klein–Gordon oscillator and the Coulomb potential," Annals Phys. **370**, 128-136 (2016). [arXiv:1511.05072 [quant-ph]].

[15] H.S. Vieira, “Resonant frequencies of the hydrodynamic vortex,” Int. J. Mod. Phys. D **26**, no.04, 1750035 (2016). [arXiv:1510.08298 [gr-qc]].

[16] Q. Dong, G.H. Sun, M.A. Aoki, C.Y. Chen and S.H. Dong, “Exact solutions of a quartic potential,” Mod. Phys. Lett. A **34**, no.26, 1950208 (2019).

[17] “HeunG”, Accessed 27 September 2020, https://reference.wolfram.com/language/ref/HeunG.html.

[18] “HeunB”, Accessed 27 September 2020, https://reference.wolfram.com/language/ref/HeunB.html.

[19] “HeunT”, Accessed 27 September 2020, https://reference.wolfram.com/language/ref/HeunT.html.

[20] M. S. Cunha and H. R. Christiansen, “Confluent Heun functions in gauge theories on thick braneworlds,” Phys. Rev. D **84**, 085002 (2011). [arXiv:1109.3486 [hep-th]].

[21] J. Derezinski, “Hypergeometric type functions and their symmetries”, Ann. Henri Poincare **15**, 1569 (2014). [arXiv:1305.3113 [math.CA]].

[22] “Solving Some Second Order Linear ODEs that Admit Hypergeometric 2F1, 1F1, and 0F1 Function Solutions”, Accessed 12 September 2020, https://www.maplesoft.com/support/help/Maple/view.aspx?path=dsolve/hyper3.

[23] “Hypergeometric2F1”, Accessed 12 September 2020,

https://functions.wolfram.com/HypergeometricFunctions/Hypergeometric2F1/13/01/01/01/.

[24] “HypergeometricU”, Accessed 12 September 2020,

https://reference.wolfram.com/language/ref/HypergeometricU.html.

[25] “The five Second Order Linear Heun equations and the corresponding Heun function solutions”, Accessed 12 September 2020, https://www.maplesoft.com/support/help/Maple/view.aspx?path=Heun.

[26] “Heun and Related Functions”, Accessed 12 September 2020,

https://reference.wolfram.com/language/guide/HeunAndRelatedFunctions.html